
2.6 Singular Value Decomposition

There exists a very powerful set of techniques for dealing with sets of equations
or matrices that are either singular or else numerically very close to singular. In many
cases where Gaussian elimination andLU decomposition fail to give satisfactory
results, this set of techniques, known assingular value decomposition, or SVD,
will diagnose for you precisely what the problem is. In some cases, SVD will
not only diagnose the problem, it will also solve it, in the sense of giving you a
useful numerical answer, although, as we shall see, not necessarily “the” answer
that you thought you should get.

SVD is also the method of choice for solving mostlinear least-squares problems.
We will outline the relevant theory in this section, but defer detailed discussion of
the use of SVD in this application to Chapter 15, whose subject is the parametric
modeling of data.

SVD methods are based on the following theorem of linear algebra, whose proof
is beyond our scope: AnyM ×N matrixA whose number of rowsM is greater than
or equal to its number of columnsN , can be written as the product of anM × N
column-orthogonal matrixU, anN × N diagonal matrixW with positive or zero
elements (thesingular values), and the transpose of anN ×N orthogonal matrixV.
The various shapes of these matrices will be made clearer by the following tableau:




A




=




U




·




w1

w2

· · ·
· · ·

wN


 ·


 VT




(2.6.1)

The matricesU and V are each orthogonal in the sense that their columns are
orthonormal,

M∑
i=1

UikUin = δkn
1 ≤ k ≤ N

1 ≤ n ≤ N
(2.6.2)

N∑
j=1

VjkVjn = δkn
1 ≤ k ≤ N

1 ≤ n ≤ N
(2.6.3)

or as a tableau,


 UT




·




U




=


 VT




·


 V




=


 1




(2.6.4)

SinceV is square, it is also row-orthonormal,V · VT = 1.
The SVD decomposition can also be carried out whenM < N . In this case

the singular valueswj for j = M + 1, . . . , N are all zero, and the corresponding
columns ofU are also zero. Equation (2.6.2) then holds only fork, n ≤ M .

The decomposition (2.6.1) can always be done, no matter how singular the
matrix is, and it is “almost” unique. That is to say, it is unique up to (i) making
the same permutation of the columns ofU, elements ofW, and columns ofV (or
rows ofVT ), or (ii) forming linear combinations of any columns ofU andV whose
corresponding elements ofW happen to be exactly equal. An important consequence
of the permutation freedom is that for the caseM < N , a numerical algorithm for
the decomposition need not return zerowj ’s for j = M + 1, . . . , N ; the N − M
zero singular values can be scattered among all positionsj = 1, 2, . . . , N .

At the end of this section, we give a routine,svdcmp, that performs SVD on
an arbitrary matrixA, replacing it byU (they are the same shape) and returning
W and V separately. The routinesvdcmp is based on a routine by Forsythe et
al. [1], which is in turn based on the original routine of Golub and Reinsch, found, in
various forms, in[2-4] and elsewhere. These references include extensive discussion
of the algorithm used. As much as we dislike the use of black-box routines, we are
going to ask you to accept this one, since it would take us too far afield to cover
its necessary background material here. Suffice it to say that the algorithm is very
stable, and that it is very unusual for it ever to misbehave. Most of the concepts that
enter the algorithm (Householder reduction to bidiagonal form, diagonalization by
QR procedure with shifts) will be discussed further in Chapter 11.

If you are as suspicious of black boxes as we are, you will want to verify yourself
thatsvdcmp does what we say it does. That is very easy to do: Generate an arbitrary
matrix A, call the routine, and then verify by matrix multiplication that (2.6.1) and
(2.6.4) are satisfied. Since these two equations are the only defining requirements
for SVD, this procedure is (for the chosenA) a complete end-to-end check.

Now let us find out what SVD is good for.

SVD of a Square Matrix

If the matrixA is square,N × N say, thenU, V, andW are all square matrices
of the same size. Their inverses are also trivial to compute:U andV are orthogonal,
so their inverses are equal to their transposes;W is diagonal, so its inverse is the
diagonal matrix whose elements are the reciprocals of the elementsw j . From (2.6.1)
it now follows immediately that the inverse ofA is

A−1 = V · [diag(1/wj)] · UT (2.6.5)

The only thing that can go wrong with this construction is for one of thew j ’s
to be zero, or (numerically) for it to be so small that its value is dominated by
roundoff error and therefore unknowable. If more than one of thew j ’s have this
problem, then the matrix is even more singular. So, first of all, SVD gives you a
clear diagnosis of the situation.

Formally, thecondition number of a matrix is defined as the ratio of the largest
(in magnitude) of thewj ’s to the smallest of thewj ’s. A matrix is singular if its
condition number is infinite, and it isill-conditioned if its condition number is too
large, that is, if its reciprocal approaches the machine’s floating-point precision (for
example, less than10−6 for single precision or10−12 for double).

For singular matrices, the concepts ofnullspace and range are important.
Consider the familiar set of simultaneous equations

A · x = b (2.6.6)

whereA is a square matrix,b andx are vectors. Equation (2.6.6) definesA as a
linear mapping from the vector spacex to the vector spaceb. If A is singular, then
there is some subspace ofx, called the nullspace, that is mapped to zero,A · x = 0.
The dimension of the nullspace (the number of linearly independent vectorsx that
can be found in it) is called thenullity of A.

Now, there is also some subspace ofb that can be “reached” byA, in the sense
that there exists somex which is mapped there. This subspace ofb is called the range
of A. The dimension of the range is called therank of A. If A is nonsingular, then its
range will be all of the vector spaceb, so its rank isN . If A is singular, then the rank
will be less thanN . In fact, the relevant theorem is “rank plus nullity equalsN .”

What has this to do with SVD? SVD explicitly constructs orthonormal bases
for the nullspace and range of a matrix. Specifically, the columns ofU whose
same-numbered elementswj arenonzero are an orthonormal set of basis vectors that
span the range; the columns ofV whose same-numbered elementsw j arezero are
an orthonormal basis for the nullspace.

Now let’s have another look at solving the set of simultaneous linear equations
(2.6.6) in the case thatA is singular. First, the set ofhomogeneous equations, where
b = 0, is solved immediately by SVD: Any column ofV whose correspondingw j

is zero yields a solution.
When the vectorb on the right-hand side is not zero, the important question is

whether it lies in the range ofA or not. If it does, then the singular set of equations
does have a solutionx; in fact it has more than one solution, since any vector in
the nullspace (any column ofV with a corresponding zerow j) can be added tox
in any linear combination.

If we want to single out one particular member of this solution-set of vectors as
a representative, we might want to pick the one with the smallest length|x| 2. Here is
how to find that vector using SVD: Simplyreplace 1/wj by zero if wj = 0. (It is not
very often that one gets to set∞ = 0 !) Then compute (working from right to left)

x = V · [diag(1/wj)] · (UT · b) (2.6.7)

This will be the solution vector of smallest length; the columns ofV that are in the
nullspace complete the specification of the solution set.

Proof: Consider|x + x′|, wherex′ lies in the nullspace. Then, ifW−1 denotes
the modified inverse ofW with some elements zeroed,

|x + x′| =
∣∣V · W−1 · UT · b + x′∣∣

=
∣∣V · (W−1 · UT · b + VT · x′)

∣∣

=
∣∣W−1 · UT · b + VT · x′

∣∣
(2.6.8)

Here the first equality follows from (2.6.7), the second and third from the orthonor-
mality of V. If you now examine the two terms that make up the sum on the
right-hand side, you will see that the first one has nonzeroj components only where
wj �= 0, while the second one, sincex′ is in the nullspace, has nonzeroj components
only wherewj = 0. Therefore the minimum length obtains forx ′ = 0, q.e.d.

If b is not in the range of the singular matrixA, then the set of equations (2.6.6)
has no solution. But here is some good news: Ifb is not in the range ofA, then
equation (2.6.7) can still be used to construct a “solution” vectorx. This vectorx
will not exactly solveA · x = b. But, among all possible vectorsx, it will do the
closest possible job in the least squares sense. In other words (2.6.7) finds

x which minimizes r ≡ |A · x − b| (2.6.9)

The numberr is called theresidual of the solution.
The proof is similar to (2.6.8): Suppose we modifyx by adding some arbitrary

x′. ThenA · x − b is modified by adding someb ′ ≡ A · x′. Obviouslyb′ is in
the range ofA. We then have

∣∣A · x − b + b′∣∣ =
∣∣(U · W · VT ) · (V · W−1 · UT · b) − b + b′∣∣

=
∣∣(U · W · W−1 · UT − 1) · b + b′∣∣

=
∣∣U · [(W · W−1 − 1) · UT · b + UT · b′]∣∣

=
∣∣(W · W−1 − 1) · UT · b + UT · b′∣∣

(2.6.10)

Now, (W · W−1 − 1) is a diagonal matrix which has nonzeroj components only for
wj = 0, while UT b′ has nonzeroj components only forwj �= 0, sinceb′ lies in the
range ofA. Therefore the minimum obtains forb ′ = 0, q.e.d.

Figure 2.6.1 summarizes our discussion of SVD thus far.

A ⋅ x = b

SVD “solution”
of A ⋅ x = c

solutions of
A ⋅ x = c′solutions of

A ⋅ x = d

null
space
of A

SVD solution of
A ⋅ x = d

range of A

d
c

(b)

(a)

A

x b

c′

Figure 2.6.1. (a) A nonsingular matrix A maps a vector space into one of the same dimension. The
vector x is mapped into b, so that x satisfies the equation A · x = b. (b) A singular matrix A maps a
vector space into one of lower dimensionality, here a plane into a line, called the “ range” of A. The
“nullspace” of A is mapped to zero. The solutions of A · x = d consist of any one particular solution plus
any vector in the nullspace, here forming a line parallel to the nullspace. Singular value decomposition
(SVD) selects the particular solution closest to zero, as shown. The point c lies outside of the range
of A, so A · x = c has no solution. SVD finds the least-squares best compromise solution, namely a
solution of A · x = c′, as shown.

In the discussion since equation (2.6.6), we have been pretending that a matrix
either is singular or else isn’ t. That is of course true analytically. Numerically,
however, the far more common situation is that some of the w j ’s are very small
but nonzero, so that the matrix is ill-conditioned. In that case, the direct solution
methods of LU decomposition or Gaussian elimination may actually give a formal
solution to the set of equations (that is, a zero pivot may not be encountered); but
the solution vector may have wildly large components whose algebraic cancellation,
when multiplying by the matrix A, may give a very poor approximation to the
right-hand vector b. In such cases, the solution vector x obtained by zeroing the

small wj ’s and then using equation (2.6.7) is very often better (in the sense of the
residual |A · x − b| being smaller) than both the direct-method solution and the SVD
solution where the small wj ’s are left nonzero.

It may seem paradoxical that this can be so, since zeroing a singular value
corresponds to throwing away one linear combination of the set of equations that
we are trying to solve. The resolution of the paradox is that we are throwing away
precisely a combination of equations that is so corrupted by roundoff error as to be at
best useless; usually it is worse than useless since it “pulls” the solution vector way
off towards infinity along some direction that is almost a nullspace vector. In doing
this, it compounds the roundoff problem and makes the residual |A · x − b| larger.

SVD cannot be applied blindly, then. You have to exercise some discretion in
deciding at what threshold to zero the small wj ’s, and/or you have to have some idea
what size of computed residual |A · x − b| is acceptable.

As an example, here is a “backsubstitution” routine svbksb for evaluating
equation (2.6.7) and obtaining a solution vector x from a right-hand side b, given
that the SVD of a matrix A has already been calculated by a call to svdcmp. Note
that this routine presumes that you have already zeroed the small w j ’s. It does not
do this for you. If you haven’t zeroed the small w j ’s, then this routine is just as
ill-conditioned as any direct method, and you are misusing SVD.

SUBROUTINE svbksb(u,w,v,m,n,mp,np,b,x)
INTEGER m,mp,n,np,NMAX
REAL b(mp),u(mp,np),v(np,np),w(np),x(np)
PARAMETER (NMAX=500) Maximum anticipated value of n.

Solves A · X = B for a vector X, where A is specified by the arrays u, w, v as returned by
svdcmp. m and n are the logical dimensions of a, and will be equal for square matrices. mp
and np are the physical dimensions of a. b(1:m) is the input right-hand side. x(1:n) is
the output solution vector. No input quantities are destroyed, so the routine may be called
sequentially with different b’s.

INTEGER i,j,jj
REAL s,tmp(NMAX)
do 12 j=1,n Calculate UT B.

s=0.
if(w(j).ne.0.)then Nonzero result only if wj is nonzero.

do 11 i=1,m
s=s+u(i,j)*b(i)

enddo 11

s=s/w(j) This is the divide by wj .
endif
tmp(j)=s

enddo 12

do 14 j=1,n Matrix multiply by V to get answer.
s=0.
do 13 jj=1,n

s=s+v(j,jj)*tmp(jj)
enddo 13

x(j)=s
enddo 14

return
END

Note that a typical use of svdcmp and svbksb superficially resembles the
typical use of ludcmp and lubksb: In both cases, you decompose the left-hand
matrix A just once, and then can use the decomposition either once or many times
with different right-hand sides. The crucial difference is the “editing” of the singular

values before svbksb is called:

REAL a(np,np),u(np,np),w(np),v(np,np),b(np),x(np)
...
do 12 i=1,n Copy a into u if you don’t want it to be destroyed.

do 11 j=1,n
u(i,j)=a(i,j)

enddo 11

enddo 12

call svdcmp(u,n,n,np,np,w,v) SVD the square matrix a.
wmax=0. Will be the maximum singular value obtained.
do 13 j=1,n

if(w(j).gt.wmax)wmax=w(j)
enddo 13

wmin=wmax*1.0e-6 This is where we set the threshold for singular values
allowed to be nonzero. The constant is typical,
but not universal. You have to experiment with
your own application.

do 14 j=1,n
if(w(j).lt.wmin)w(j)=0.

enddo 14

call svbksb(u,w,v,n,n,np,np,b,x) Now we can backsubstitute.

SVD for Fewer Equations than Unknowns

If you have fewer linear equations M than unknowns N , then you are not
expecting a unique solution. Usually there will be an N − M dimensional family
of solutions. If you want to find this whole solution space, then SVD can readily
do the job.

The SVD decomposition will yield N − M zero or negligible wj ’s, since
M < N . There may be additional zero wj ’s from any degeneracies in your M
equations. Be sure that you find this many small wj ’s, and zero them before calling
svbksb, which will give you the particular solution vector x. As before, the columns
of V corresponding to zeroed wj ’s are the basis vectors whose linear combinations,
added to the particular solution, span the solution space.

SVD for More Equations than Unknowns

This situation will occur in Chapter 15, when we wish to find the least-squares
solution to an overdetermined set of linear equations. In tableau, the equations
to be solved are




A




·


x


 =




b




(2.6.11)

The proofs that we gave above for the square case apply without modification
to the case of more equations than unknowns. The least-squares solution vector x is

given by (2.6.7), which, with nonsquare matrices, looks like this,


x


 =


 V


 ·


diag(1/wj)


 ·


 UT


 ·




b




(2.6.12)

In general, the matrix W will not be singular, and no w j ’s will need to be
set to zero. Occasionally, however, there might be column degeneracies in A. In
this case you will need to zero some small wj values after all. The corresponding
column in V gives the linear combination of x’s that is then ill-determined even by
the supposedly overdetermined set.

Sometimes, although you do not need to zero any w j ’s for computational
reasons, you may nevertheless want to take note of any that are unusually small:
Their correspondingcolumns in V are linear combinations of x’s which are insensitive
to your data. In fact, you may then wish to zero these w j ’s, to reduce the number of
free parameters in the fit. These matters are discussed more fully in Chapter 15.

Constructing an Orthonormal Basis

Suppose that you have N vectors in an M -dimensional vector space, with
N ≤ M . Then the N vectors span some subspace of the full vector space.
Often you want to construct an orthonormal set of N vectors that span the same
subspace. The textbook way to do this is by Gram-Schmidt orthogonalization,
starting with one vector and then expanding the subspace one dimension at a
time. Numerically, however, because of the build-up of roundoff errors, naive
Gram-Schmidt orthogonalization is terrible.

The right way to construct an orthonormal basis for a subspace is by SVD:
Form an M × N matrix A whose N columns are your vectors. Run the matrix
through svdcmp. The columns of the matrix U (which in fact replaces A on output
from svdcmp) are your desired orthonormal basis vectors.

You might also want to check the output wj ’s for zero values. If any occur,
then the spanned subspace was not, in fact, N dimensional; the columns of U
corresponding to zero wj ’s should be discarded from the orthonormal basis set.

(QR factorization, discussed in §2.10, also constructs an orthonormal basis,
see [5].)

Approximation of Matrices

Note that equation (2.6.1) can be rewritten to express any matrix A ij as a sum
of outer products of columns of U and rows of V T , with the “weighting factors”
being the singular values wj ,

Aij =
N∑

k=1

wk UikVjk (2.6.13)

If you ever encounter a situation where most of the singular values w j of a
matrix A are very small, then A will be well-approximated by only a few terms in the
sum (2.6.13). This means that you have to store only a few columns of U and V (the
same k ones) and you will be able to recover, with good accuracy, the whole matrix.

Note also that it is very efficient to multiply such an approximated matrix by a
vector x: You just dot x with each of the stored columns of V, multiply the resulting
scalar by the corresponding wk, and accumulate that multiple of the corresponding
column of U. If your matrix is approximated by a small number K of singular
values, then this computation of A · x takes only about K(M + N) multiplications,
instead of MN for the full matrix.

SVD Algorithm

Here is the algorithm for constructing the singular value decomposition of any
matrix. See §11.2–§11.3, and also [4-5], for discussion relating to the underlying
method.

SUBROUTINE svdcmp(a,m,n,mp,np,w,v)
INTEGER m,mp,n,np,NMAX
REAL a(mp,np),v(np,np),w(np)
PARAMETER (NMAX=500) Maximum anticipated value of n.

C USES pythag
Given a matrix a(1:m,1:n), with physical dimensions mp by np, this routine computes its
singular value decomposition, A = U · W · V T . The matrix U replaces a on output. The
diagonal matrix of singular values W is output as a vector w(1:n). The matrix V (not the
transpose V T ) is output as v(1:n,1:n).

INTEGER i,its,j,jj,k,l,nm
REAL anorm,c,f,g,h,s,scale,x,y,z,rv1(NMAX),pythag
g=0.0 Householder reduction to bidiagonal form.
scale=0.0
anorm=0.0
do 25 i=1,n

l=i+1
rv1(i)=scale*g
g=0.0
s=0.0
scale=0.0
if(i.le.m)then

do 11 k=i,m
scale=scale+abs(a(k,i))

enddo 11

if(scale.ne.0.0)then
do 12 k=i,m

a(k,i)=a(k,i)/scale
s=s+a(k,i)*a(k,i)

enddo 12

f=a(i,i)
g=-sign(sqrt(s),f)
h=f*g-s
a(i,i)=f-g
do 15 j=l,n

s=0.0
do 13 k=i,m

s=s+a(k,i)*a(k,j)
enddo 13

f=s/h
do 14 k=i,m

a(k,j)=a(k,j)+f*a(k,i)
enddo 14

enddo 15

do 16 k=i,m
a(k,i)=scale*a(k,i)

enddo 16

endif
endif
w(i)=scale *g
g=0.0
s=0.0
scale=0.0
if((i.le.m).and.(i.ne.n))then

do 17 k=l,n
scale=scale+abs(a(i,k))

enddo 17

if(scale.ne.0.0)then
do 18 k=l,n

a(i,k)=a(i,k)/scale
s=s+a(i,k)*a(i,k)

enddo 18

f=a(i,l)
g=-sign(sqrt(s),f)
h=f*g-s
a(i,l)=f-g
do 19 k=l,n

rv1(k)=a(i,k)/h
enddo 19

do 23 j=l,m
s=0.0
do 21 k=l,n

s=s+a(j,k)*a(i,k)
enddo 21

do 22 k=l,n
a(j,k)=a(j,k)+s*rv1(k)

enddo 22

enddo 23

do 24 k=l,n
a(i,k)=scale*a(i,k)

enddo 24

endif
endif
anorm=max(anorm,(abs(w(i))+abs(rv1(i))))

enddo 25

do 32 i=n,1,-1 Accumulation of right-hand transformations.
if(i.lt.n)then

if(g.ne.0.0)then
do 26 j=l,n Double division to avoid possible underflow.

v(j,i)=(a(i,j)/a(i,l))/g
enddo 26

do 29 j=l,n
s=0.0
do 27 k=l,n

s=s+a(i,k)*v(k,j)
enddo 27

do 28 k=l,n
v(k,j)=v(k,j)+s*v(k,i)

enddo 28

enddo 29

endif
do 31 j=l,n

v(i,j)=0.0
v(j,i)=0.0

enddo 31

endif
v(i,i)=1.0

g=rv1(i)
l=i

enddo 32

do 39 i=min(m,n),1,-1 Accumulation of left-hand transformations.
l=i+1
g=w(i)
do 33 j=l,n

a(i,j)=0.0
enddo 33

if(g.ne.0.0)then
g=1.0/g
do 36 j=l,n

s=0.0
do 34 k=l,m

s=s+a(k,i)*a(k,j)
enddo 34

f=(s/a(i,i))*g
do 35 k=i,m

a(k,j)=a(k,j)+f*a(k,i)
enddo 35

enddo 36

do 37 j=i,m
a(j,i)=a(j,i)*g

enddo 37

else
do 38 j= i,m

a(j,i)=0.0
enddo 38

endif
a(i,i)=a(i,i)+1.0

enddo 39

do 49 k=n,1,-1 Diagonalization of the bidiagonal form: Loop over
singular values, and over allowed iterations.do 48 its=1,30

do 41 l=k,1,-1 Test for splitting.
nm=l-1 Note that rv1(1) is always zero.
if((abs(rv1(l))+anorm).eq.anorm) goto 2
if((abs(w(nm))+anorm).eq.anorm) goto 1

enddo 41

1 c=0.0 Cancellation of rv1(l), if l > 1.
s=1.0
do 43 i=l,k

f=s*rv1(i)
rv1(i)=c*rv1(i)
if((abs(f)+anorm).eq.anorm) goto 2
g=w(i)
h=pythag(f,g)
w(i)=h
h=1.0/h
c= (g*h)
s=-(f*h)
do 42 j=1,m

y=a(j,nm)
z=a(j,i)
a(j,nm)=(y*c)+(z*s)
a(j,i)=-(y*s)+(z*c)

enddo 42

enddo 43

2 z=w(k)
if(l.eq.k)then Convergence.

if(z.lt.0.0)then Singular value is made nonnegative.
w(k)=-z
do 44 j=1,n

v(j,k)=-v(j,k)
enddo 44

endif
goto 3

endif
if(its.eq.30) pause ’no convergence in svdcmp’
x=w(l) Shift from bottom 2-by-2 minor.
nm=k-1
y=w(nm)
g=rv1(nm)
h=rv1(k)
f=((y-z)*(y+z)+(g-h)*(g+h))/(2.0*h*y)
g=pythag(f,1.0)
f=((x-z)*(x+z)+h*((y/(f+sign(g,f)))-h))/x
c=1.0 Next QR transformation:
s=1.0
do 47 j=l,nm

i=j+1
g=rv1(i)
y=w(i)
h=s*g
g=c*g
z=pythag(f,h)
rv1(j)=z
c=f/z
s=h/z
f= (x*c)+(g*s)
g=-(x*s)+(g*c)
h=y*s
y=y*c
do 45 jj=1,n

x=v(jj,j)
z=v(jj,i)
v(jj,j)= (x*c)+(z*s)
v(jj,i)=-(x*s)+(z*c)

enddo 45

z=pythag(f,h)
w(j)=z Rotation can be arbitrary if z = 0.
if(z.ne.0.0)then

z=1.0/z
c=f*z
s=h*z

endif
f= (c*g)+(s*y)
x=-(s*g)+(c*y)
do 46 jj=1,m

y=a(jj,j)
z=a(jj,i)
a(jj,j)= (y*c)+(z*s)
a(jj,i)=-(y*s)+(z*c)

enddo 46

enddo 47

rv1(l)=0.0
rv1(k)=f
w(k)=x

enddo 48

3 continue
enddo 49

return
END

FUNCTION pythag(a,b)
REAL a,b,pythag

Computes (a2 + b2)1/2 without destructive underflow or overflow.

REAL absa,absb
absa=abs(a)
absb=abs(b)
if(absa.gt.absb)then

pythag=absa*sqrt(1.+(absb/absa)**2)
else

if(absb.eq.0.)then
pythag=0.

else
pythag=absb*sqrt(1.+(absa/absb)**2)

endif
endif
return
END

(Double precision versions of svdcmp, svbksb, and pythag, named dsvdcmp,
dsvbksb, and dpythag, are used by the routine ratlsq in §5.13. You can easily
make the conversions, or else get the converted routines from the Numerical Recipes
diskette.)

2.7 Sparse Linear Systems

A system of linear equations is called sparse if only a relatively small number
of its matrix elements aij are nonzero. It is wasteful to use general methods of
linear algebra on such problems, because most of the O(N 3) arithmetic operations
devoted to solving the set of equations or inverting the matrix involve zero operands.
Furthermore, you might wish to work problems so large as to tax your available
memory space, and it is wasteful to reserve storage for unfruitful zero elements.
Note that there are two distinct (and not always compatible) goals for any sparse
matrix method: saving time and/or saving space.

We have already considered one archetypal sparse form in §2.4, the band
diagonal matrix. In the tridiagonal case, e.g., we saw that it was possible to save


